
NPN_Ftau_15uu_XTF 
*
*
*
*                            ________
*                           |        |
*                           |       _|_
*                           |C     /VCC\
*             R1           _|      \___/
*      ____/\  /\  /\____|'QN1       |
*    _|_     \/  \/   B  |`->       _|_
*    ///                    |E      ///
*                           |
*                          _|_
*                         / _ \
*                      1I \/ \/
*                         /\_/\
*                         \___/
*                          _|_
*                          ///
.OPTIONS    GMIN=1e-15   METHOD=gear   ABSTOL=1e-15  TEMP=27 srcsteps = 1  gminsteps = 1    
VCC    C    0     DC     5V     
QN1    C    B     E      NPNV
R1     0    B            1
I1     E    0     DC     100u    AC    .01u
.control
echo             "15um^2 Ftau vs IE and CJE"
setplot           new  
set NameList =    ( XTF_1m XTF_1 XTF_10 )
compose XtfVals   values  1m 1 10 
compose IbiasVals values 10u 100u 1m 10m 30m 60m 100m
settype current   IbiasVals
let NoOfXtf  =   length(XtfVals)
let NoOfIbias =   length(IbiasVals)
begin
unset             interrupt
* ================Loop_CJE============
let j =           1
while             ( j <= NoOfXtf )
let xtf =         XtfVals[j-1]
altermod          npnv xtf = $&xtf 
echo              "=>Ibias = Ftau@XTF=$&xtf"
set thisName =    $NameList[$&j]
let $thisName =   0*vector(NoOfIbias)
* ================Loop_Ibias==============
let k =           1
while             ( k <= NoOfIbias )
let Ibias =       IbiasVals[k-1]
alter I1 dc =     $&Ibias
alter I1 ac =     .01u
ac                dec 30 .0001GHz 100GHz
* ================Find_Ftau============= Use
let g =           (-db(b) - 160)
let f =           3.16*mag(frequency[ sortorder( abs(g-10.0) )[0] ] )
echo              "$&unknown.Ibias   $&f"
let               unknown.{$thisName}[unknown.k-1] = f
destroy
if                ($?interrupt)
goto              bail
endif
let k =           k + 1
endwhile
let j =           j + 1
endwhile
setscale          IbiasVals
plot              $NameList loglog title "15um^2 Ftau vs IE and XTF"
label             bail
echo             "Done."
end
.endc



.MODEL  NPNV      NPN(  
+ IS=10.15e-18    NF=1.0       BF=89        VAF=107     IKF=.4  
+                 NR=1.006     BR=0.4822    VAR=4.286   IKR=0.0002472
+ ISE=9.15E-17    NE=2                                 
+ ISC=1E-21       NC=2                                                 
+ RB=12           RBM=3                                 IRB=.1 
+ RE=2            RC=33                                         
+ CJE=37e-15      VJE=0.75     MJE=0.35                            
+ CJC=45E-15      VJC=0.6399   MJC=0.3531                         
+ CJS=2.939E-15   VJS=0.3488   MJS=0.1813  XCJC=0.4201                 
+ TF=1e-12        XTF=1        VTF=1.5     ITF=.8         
+ TR=5.9e-9       FC=0.5       PTF=30                                 
+ KF=1.000E-16    AF=1      
+ XTB=2           EG=1.11      XTI=5       TNOM=25       )

.end
=====================END_OF_SPICE=======================

**
**                                ___
**                               |(C)|    NPN
**     SPICE MODEL               |___|
**                                 |
**                                 / RC 33ohms    VAF=107
**                                 \
**                                 /                     3fF
**         ________________________|_____________________
**        |       |      |      |    |      |     _|_   _|_Cjs
**        |       |45fF  |      |    |Ir/BR |Irn / _ \  ___      
**        |       |      |      /   _|_    _|_   \/ \/   |gnd!
**       _|_Cjcx _|_Cjc _|_Cdc  \    ^      ^    /\_/\  _|_
**       ___     ___    ___     /   /_\    /_\   \___/ \sub/
**   ___  |   Rbb'|      | gmin \    |      |   Ic |    \ /
**  |(B)|_|_/\  __|______|______|____|______|      | |   V
**  |___|     \/  |      |      |   _|_    _|_     | V (hidden)
**      12ohms   _|_Cje _|_Cde  \   \ /    \ /     |
**     BF=89     ___    ___     /   _v_    _v_     |
**                |37fF  | gmin \    |If/Bf |Ifn   |
**                |______|______|____|______|______|
**                                 |
**                  TF=8ps         / RE 2ohms
**                                 \
**                                 /
**                                _|_           
**                               |(E)|
**                               |___|
**



=


